
1	Decision/action requested
The group is requested to discuss and approve the pCR below

2	References
[1]		3GPP TS 28.831: " Management and orchestration; Study on basic Service-Based Management Architecture (SBMA) enabler enhancements"
3	Rationale
None.
4	Detailed proposal
The following changes are proposed for TR 28.831[1].
	Begin of modifications



4.2.5.2.4	XPath 1.0 profiles
An XPath expression matches the production "Expr" defined in XPath 1.0 [2], clause 3.1. However, the general expression is much too generic for selecting nodes of an input document. For rexample, it allows also expression like
(5, 256)[2]
which selects the second item in the sequence (5, 256), hence 256. This expression does not work at all on an input document. Even if "5" and "256" is replaced by some XPath expression evaluating to numbers, the expression cannot be used for selecting nodes.
This is why only one or more XPath profiles are needed.
Two XPath profiles are proposed: 
· [bookmark: _Hlk110584445]Basic profile: Supporting limited features allowing XPath to browse the document from one element node to another. The XPath EBNF is detailed in annex A.1. In this profile a location path is defined as an absolute location path. An absolute location path consists of a sequence of one or more location steps separated by / and preceded by /. The location steps in an absolute location path are composed together from left to right. Each step-in turn selects a set of nodes relative to a context node. Note that a / by itself selects the root node of the document containing the context node.
Example: /SubNetwork/ManagedElement/attributes

A location step is composed of
· an axis specifier, which specifies the tree relationship between the nodes selected by the step and the context node
· a node name which specifies the node name of the node selected by the location step

The axis specifier includes 2 axis: 
· Child: axis containing the children of the context node
Example: /SubNetwork /child::*
· Descendant: axis containing the descendants of the context node; a descendant is a child or a child of a child and so on
Example: /SubNetwork/descendant::*
· [bookmark: _Hlk110594337][bookmark: _Hlk110592969][bookmark: _Hlk110593433]Advanced profile: supporting more advanced features like the usage of the predicates. The XPath EBNF is detailed in annex A.2.  In this profile, a relative location path is added as a second option to the location path. A relative location path consists of a sequence of one or more location steps separated by / and it does not need to start from the root node as the absolute path.
Example: / / attributes
In this profile, in addition to the axis specifier and Node Name (same as the basic profile), a predicate option is added to the location step. The predicate uses arbitrary expressions to further refine the set of nodes selected by the step. Predicates are defined by an expression (Expr) always embedded in square brackets. An Expr can be as follows: 
· An OR expression: evaluated by evaluating each operand and converting its value to a Boolean. The result is true if either value is true and false otherwise. The right operand is not evaluated if the left operand evaluates to true.
· An AND expression: evaluated by evaluating each operand and converting its value to Bolean. The result is true if both values are true and false otherwise. The right operand is not evaluated if the left operand evaluates to false.
· An EqualityExpr: evaluated by comparing the objects that result from evaluating the two operands. In the EqualityExpr both objects to be compared are node-sets. More details on handling comparison depending on the node-set types are given in clause 3.4 [2].
· RelationalExpr defines a comparison of a node-set to a given number. The object to be compared here is converted to a number as if by applying the number function (more details in clause 3.5 [2]).
Examples: 
//attributes[granularityPeriod  != 4 ]
//attributes[location = 'TV Tower']
//attributes[granularityPeriod > 4 or granularityPeriod < 1 ]
In this profile, a FunctionCall is defined. It gives access to the XPath function librairy used to evaluates expressions. Each function in the function library is specified using a function prototype, which gives the return type, the name of the function, and the type of the arguments (more details in Clause 4[2]).  
Examples:
//ManagedElement[position()=1] 
/SubNetwork/*[count(/SubNetwork/*)]
//ManagedElement[starts-with(id,"ME2")]
//ManagedElement[contains(id,"ME")]
//ManagedElement[contains(id,substring-before(//ManagedElement[1]/id, '1'))]

The rest of the Grammar introduced defines additional rules for the lexical structure (clause 3.7[2]).

Note that a context node for the evaluation of the XPath expression is the base object, i.e. the object that name-contains the object with the XPath expression.

Annex A 
A.1 EBNF for basic XPath profile 
LocationPath				::= AbsoluteLocationPath
AbsoluteLocationPath	::= Step
									| AbsoluteLocationPath '/' Step
Step								::= AxisSpecifier NodeName
AxisSpecifier				::= AxisName '::'
AxisName						::= 'child'
| 'descendant'       
NodeName						::= '*'
| NCName ':' '*'
| QName

A.2 EBNF for advanced XPath profile 
LocationPath		::= RelativeLocationPath
| AbsoluteLocationPath
AbsoluteLocationPath
::= '/' RelativeLocationPath?
| '//' RelativeLocationPath
RelativeLocationPath
::= Step
| RelativeLocationPath '/' Step
| RelativeLocationPath '//' Step
Step						::= AxisSpecifier NodeName Predicate*
AxisSpecifier		::= AxisName '::'
AxisName				::= 'child'
| 'descendant'
NodeName 				::= '*'
| NCName ':' '*'
| QName
Predicate				::= '[' PredicateExpr ']'
PredicateExpr 		::= Expr
Expr						::= OrExpr
OrExpr					::= AndExpr
| OrExpr 'or' AndExpr
AndExpr					::= EqualityExpr
| AndExpr 'and' EqualityExpr
EqualityExpr		::= RelationalExpr
| EqualityExpr '=' RelationalExpr
| EqualityExpr '!=' RelationalExpr
RelationalExpr		::= UnaryExpr
| RelationalExpr '<' Number
| RelationalExpr '>' Number
| RelationalExpr '<=' Number
| RelationalExpr '>=' Number
UnaryExpr				::= UnionExpr
| '-' UnaryExpr
UnionExpr				::= PathExpr
PathExpr				::= LocationPath
| FilterExpr
| FilterExpr '/' RelativeLocationPath
| FilterExpr '//' RelativeLocationPath
FilterExpr			::= PrimaryExpr
| FilterExpr Predicate
PrimaryExpr			::= VariableReference
| Literal
| Number
| FunctionCall
FunctionCall		::= FunctionName '(' ( Argument ( ',' Argument )* )? ')'
Argument 				::= Expr
ExprToken				::= '(' | ')' | '[' | ']' | '.' | '..' | ',' | '::'
| NameTest
| Operator
| FunctionName
| AxisName
| Literal
| Number
| VariableReference
Literal					::= '"' [^"]* '"'
| "'" [^']* "'"
Number					::= Digits ('.' Digits?)?
| '.' Digits
Digits					::= [0-9]+
Operator				::= OperatorName
| '/' | '//' | '|' | '+' | '-' | '=' | '!=' | '<' | '<=' | '>' | '>='
OperatorName		::= 'and' | 'or' 
FunctionName		::= QName - NodeType
VariableReference::= '$' QName
[bookmark: _Hlk110590033]ExprWhitespace		 ::= S

